# Example 6 -- The coordinate ring of the spinor variety

Consider the complex Lie group $SO(10)$ of type $D_5$. We denote $V(\omega)$ the highest weight representation of $SO(10)$ with highest weight $\omega$. We denote by $\omega_1,...,\omega_5$ the fundamental weights in the root system of type $D_5$.

We will obtain the minimal free resolution of the coordinate ring of the spinor variety of type $D_5$ (see Rincon - Isotropical linear spaces and valuated delta-matroids, Sec. 2, for a concise introduction to spinor varieties). The affine cone over the spinor variety of type $D_5$ lives in the representation $V(\omega_5)$, the 5th fundamental representation of $SO(10)$, considered as an affine space. Polynomial functions on this affine space are given by the symmetric algebra over the dual representation, i.e., $V(\omega_4)$.

Following the description in Fulton, Harris - Representation Theory, Ch. 20.1, we can construct $V(\omega_4)$ as $\wedge^0 E \oplus \wedge^2 E \oplus \wedge^4 E$, where $E$ is a 5 dimensional complex vector space. Let $\{e_0,...,e_4\}$ be a basis of $E$. Then a basis of $V(\omega_4)$ is given by the exterior products $e_J = e_{j_1} \wedge ... \wedge e_{j_{2r}}$, for all subsets $J=\{j_1,...,j_{2r}\}$ of even cardinality of $\{0,..., 4\}$. Denote by $x_J$ the variable corresponding to $e_J$ in $R$.

The spinor variety of type $D_5$ is cut out by quadratic equations which represent all possible relations among the sub Pfaffians of a $5\times 5$ generic skew symmetric matrix. A general description can be found for example in Manivel - On Spinor Varieties and Their Secants.

 i1 : R=QQ[x_{}, x_{0,1}, x_{0,2}, x_{1,2}, x_{0,3}, x_{1,3}, x_{2,3}, x_{0,4}, x_{1,4}, x_{2,4}, x_{3,4}, x_{0,1,2,3}, x_{0,1,2,4}, x_{0,1,3,4}, x_{0,2,3,4}, x_{1,2,3,4}] o1 = R o1 : PolynomialRing i2 : I=ideal(x_{}*x_{0,1,2,3}-x_{0,1}*x_{2,3}+x_{0,2}*x_{1,3}-x_{0,3}*x_{1,2}, x_{}*x_{0,1,2,4}-x_{0,1}*x_{2,4}+x_{0,2}*x_{1,4}-x_{0,4}*x_{1,2}, x_{}*x_{0,1,3,4}-x_{0,1}*x_{3,4}+x_{0,3}*x_{1,4}-x_{0,4}*x_{1,3}, x_{}*x_{0,2,3,4}-x_{0,2}*x_{3,4}+x_{0,3}*x_{2,4}-x_{0,4}*x_{2,3}, x_{}*x_{1,2,3,4}-x_{1,2}*x_{3,4}+x_{1,3}*x_{2,4}-x_{1,4}*x_{2,3}, x_{0,1}*x_{0,2,3,4}-x_{0,2}*x_{0,1,3,4}+x_{0,3}*x_{0,1,2,4}-x_{0,4}*x_{0,1,2,3}, -x_{0,1}*x_{1,2,3,4}+x_{1,2}*x_{0,1,3,4}-x_{1,3}*x_{0,1,2,4}+x_{1,4}*x_{0,1,2,3}, x_{0,2}*x_{1,2,3,4}-x_{1,2}*x_{0,2,3,4}+x_{2,3}*x_{0,1,2,4}-x_{2,4}*x_{0,1,2,3}, -x_{0,3}*x_{1,2,3,4}+x_{1,3}*x_{0,2,3,4}-x_{2,3}*x_{0,1,3,4}+x_{3,4}*x_{0,1,2,3}, x_{0,4}*x_{1,2,3,4}-x_{1,4}*x_{0,2,3,4}+x_{2,4}*x_{0,1,3,4}-x_{3,4}*x_{0,1,2,4}); o2 : Ideal of R i3 : RI=res I; betti RI 0 1 2 3 4 5 o4 = total: 1 10 16 16 10 1 0: 1 . . . . . 1: . 10 16 . . . 2: . . . 16 10 . 3: . . . . . 1 o4 : BettiTally

The root system of type $D_5$ is contained in $\RR^5$. It is easy to express the weight of each variable of the ring $R$ with respect to the coordinate basis of $\RR^5$. The weight of $x_J$ is a vector $(a_1,...,a_5)\in\RR^5$, with $a_k = 1/2$ if $k\in J$ and $a_k = -1/2$ otherwise.

 i5 : ind = apply(gens R,g->(baseName g)#1) o5 = {{}, {0, 1}, {0, 2}, {1, 2}, {0, 3}, {1, 3}, {2, 3}, {0, 4}, {1, 4}, {2, ------------------------------------------------------------------------ 4}, {3, 4}, {0, 1, 2, 3}, {0, 1, 2, 4}, {0, 1, 3, 4}, {0, 2, 3, 4}, {1, ------------------------------------------------------------------------ 2, 3, 4}} o5 : List i6 : makeWeight = J -> apply(5,i->if member(i,J) then 1/2 else -1/2) o6 = makeWeight o6 : FunctionClosure i7 : W'=apply(ind,makeWeight) 1 1 1 1 1 1 1 1 1 1 1 1 1 1 o7 = {{- -, - -, - -, - -, - -}, {-, -, - -, - -, - -}, {-, - -, -, - -, - 2 2 2 2 2 2 2 2 2 2 2 2 2 2 ------------------------------------------------------------------------ 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 -}, {- -, -, -, - -, - -}, {-, - -, - -, -, - -}, {- -, -, - -, -, - -}, 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 ------------------------------------------------------------------------ 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 {- -, - -, -, -, - -}, {-, - -, - -, - -, -}, {- -, -, - -, - -, -}, {- 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 ------------------------------------------------------------------------ 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 -, - -, -, - -, -}, {- -, - -, - -, -, -}, {-, -, -, -, - -}, {-, -, -, 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 ------------------------------------------------------------------------ 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 - -, -}, {-, -, - -, -, -}, {-, - -, -, -, -}, {- -, -, -, -, -}} 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 o7 : List

Now we convert these weights into the basis of fundamental weights. To achieve this we make each previous weight into a column vector and join all column vectors into a matrix. Then we multiply on the left by the matrix $M$ expressing the change of basis from the coordinate basis of $\RR^5$ to the base of simple roots of $D_5$ (as described in Humphreys - Introduction to Lie Algebras and Representation Theory, Ch. 12.1). Finally we multiply the resulting matrix on the left by $N$, the transpose of the Cartan matrix of $D_5$, which expresses the change of basis from the simple roots to the fundamental weights of $D_5$. The columns of the matrix thus obtained are the desired weights, so they can be attached to the ring $R$.

 i8 : M=inverse promote(matrix{{1,0,0,0,0},{-1,1,0,0,0},{0,-1,1,0,0},{0,0,-1,1,1},{0,0,0,-1,1}},QQ) o8 = | 1 0 0 0 0 | | 1 1 0 0 0 | | 1 1 1 0 0 | | 1/2 1/2 1/2 1/2 -1/2 | | 1/2 1/2 1/2 1/2 1/2 | 5 5 o8 : Matrix QQ <--- QQ i9 : D=dynkinType{{"D",5}} o9 = DynkinType{{D, 5}} o9 : DynkinType i10 : N=transpose promote(cartanMatrix(rootSystem(D)),QQ) o10 = | 2 -1 0 0 0 | | -1 2 -1 0 0 | | 0 -1 2 -1 -1 | | 0 0 -1 2 0 | | 0 0 -1 0 2 | 5 5 o10 : Matrix QQ <--- QQ i11 : W=entries transpose lift(N*M*(transpose matrix W'),ZZ) o11 = {{0, 0, 0, 0, -1}, {0, 1, 0, 0, -1}, {1, -1, 1, 0, -1}, {-1, 0, 1, 0, ----------------------------------------------------------------------- -1}, {1, 0, -1, 1, 0}, {-1, 1, -1, 1, 0}, {0, -1, 0, 1, 0}, {1, 0, 0, ----------------------------------------------------------------------- -1, 0}, {-1, 1, 0, -1, 0}, {0, -1, 1, -1, 0}, {0, 0, -1, 0, 1}, {0, 0, ----------------------------------------------------------------------- 0, 1, 0}, {0, 0, 1, -1, 0}, {0, 1, -1, 0, 1}, {1, -1, 0, 0, 1}, {-1, 0, ----------------------------------------------------------------------- 0, 0, 1}} o11 : List i12 : setWeights(R,D,W) o12 = Tally{{0, 0, 0, 1, 0} => 1} o12 : Tally

At this stage, we can issue the command to decompose the resolution.

 i13 : highestWeightsDecomposition(RI) o13 = HashTable{0 => HashTable{{0} => Tally{{0, 0, 0, 0, 0} => 1}}} 1 => HashTable{{2} => Tally{{1, 0, 0, 0, 0} => 1}} 2 => HashTable{{3} => Tally{{0, 0, 0, 0, 1} => 1}} 3 => HashTable{{5} => Tally{{0, 0, 0, 1, 0} => 1}} 4 => HashTable{{6} => Tally{{1, 0, 0, 0, 0} => 1}} 5 => HashTable{{8} => Tally{{0, 0, 0, 0, 0} => 1}} o13 : HashTable

We deduce that the resolution has the following structure $$R \leftarrow V(\omega_1) \otimes R(-2) \leftarrow V(\omega_5) \otimes R(-3) \leftarrow V(\omega_4) \otimes R(-5) \leftarrow V(\omega_1) \otimes R(-6) \leftarrow R(-8) \leftarrow 0$$

Let us also decompose some graded components of the quotient $R/I$.

 i14 : highestWeightsDecomposition(R/I,0,4) o14 = HashTable{0 => Tally{{0, 0, 0, 0, 0} => 1}} 1 => Tally{{0, 0, 0, 1, 0} => 1} 2 => Tally{{0, 0, 0, 2, 0} => 1} 3 => Tally{{0, 0, 0, 3, 0} => 1} 4 => Tally{{0, 0, 0, 4, 0} => 1} o14 : HashTable

We deduce that, for $d\in\{0,...,4\}$, $(R/I)_d = V(d\omega_4)$.